
1 Folds On Lists
A Common Pattern We’ve Seen Many Times. . .

sum [ ] = 0
sum (x : xs) = x+ sum xs

length [ ] = 0
length (x : xs) = 1 + length xs

map f [ ] = [ ]
map f (x : xs) = f x : map f xs

This pattern is extracted and called foldr :

foldr f e [ ] = e,
foldr f e (x : xs) = f x (foldr f e xs).

For easy reference, we sometimes call e the “base
value” and f the “step function.”

1.1 The Ubiquitous foldr
Replacing Constructors

foldr f e [ ] = e
foldr f e (x : xs) = f x (foldr f e xs)

• One way to look at foldr (⊕) e is that it replaces [ ]
with e and (:) with (⊕):

foldr (⊕) e [1, 2, 3, 4]
= foldr (⊕) e (1 : (2 : (3 : (4 : [ ]))))
= 1⊕ (2⊕ (3⊕ (4⊕ e))).

• sum = foldr (+) 0.

• length = foldr (λx n.1 + n) 0.

• map f = foldr (λx xs .f x : xs) [ ].

• One can see that id = foldr (:) [ ].

Some Trivial Folds on Lists

• Function max returns the maximum element in a
list:

max [ ] = -∞,
max (x : xs) = x ↑ max xs .

max = foldr (↑) -∞.

• This function is actually called maximum in the
standard Haskell Prelude, while max returns the
maximum between its two arguments. For brevity,
we denote the former bymax and the latter by (↑).

• Function prod returns the product of a list:

prod [ ] = 1,
prod (x : xs) = x× prod xs .

prod = foldr (×) 1.

• Function and returns the conjunction of a list:

and [ ] = true,
and (x : xs) = x ∧ and xs .

and = foldr (∧) true.

• Lets emphasise again that id on lists is a fold:

id [ ] = [ ],
id (x : xs) = x : id xs .

id = foldr (:) [ ].

Some Functions We Have Seen. . .

• (++ ys) = foldr (:) ys .

(++) :: List a → List a → List a
[ ] ++ ys = ys
(x : xs)++ ys = x : (xs ++ ys) .

• concat = foldr (++) [ ].

concat :: List (List a) → List a
concat [ ] = [ ]
concat (xs : xss) = xs ++ concat xss .

Replacing Constructors

• Understanding foldr from its type. Recall

data List a = [ ] | a : List a .

• Types of the two constructors: [ ] :: List a, and (:) ::
a → List a → List a.

• foldr replaces the constructors:

foldr :: (a → b → b) → b → List a → b
foldr f e [ ] = e
foldr f e (x : xs) = f x (foldr f e xs) .

Functions on Lists That Are Not foldr

• A function f is a foldr if in f (x : xs) = ...f xs..,
the argument xs does not appear outside of the re-
cursive call.

• Not all functions taking a list as input is a foldr .

• The canonical example is perhaps tail :: List a →
List a.

– tail(x : xs) = ...tail xs..??

– tail dropped too much information, which
cannot be recovered.

• Another example is dropWhile p :: List a → List a.

Longest Prefix

• The function call takeWhile p xs returns the longest
prefix of xs that satisfies p:

takeWhile p [ ] = [ ]
takeWhile p (x : xs) =
if p x then x : takeWhile p xs
else [ ] .

• E.g. takeWhile (≤ 3) [1, 2, 3, 4, 5] = [1, 2, 3].

• It can be defined by a fold:

takeWhile p
foldr (λx xs → if p x then x : xs else [ ]) [ ].

All Prefixes

• The function inits returns the list of all prefixes of
the input list:

inits [ ] = [[ ]],
inits (x : xs) = [ ] : map (x :) (inits xs).

• E.g. inits [1, 2, 3] = [[ ], [1], [1, 2], [1, 2, 3]].

• It can be defined by a fold:

inits = foldr (λx xss → [ ] : map (x :) xss) [[ ]].

All Suffixes

• The function tails returns the list of all suffixes of
the input list:

tails [ ] = [[ ]],
tails (x : xs) = (x : xs) : tails xs .

– It appears that tails is not a foldr !

• Luckily, we have head (tails xs) = xs . Therefore,

tails (x : xs) = let yss = tails xs
in (x : head yss) : yss .

• The function tails may thus be defined by a fold:

tails = foldr (λx yss →
(x : head yss) : yss) [[ ]].

1.2 The Fold-Fusion Theorem
Why Folds?

• “What are the threemost important factors in a pro-
gramming language?” Abstraction, abstraction, and
abstraction!

• Control abstraction, procedure abstraction, data
abstraction,. . . can programming patterns be ab-
stracted too?

• Program structure becomes an entity we can talk
about, reason about, and reuse.

– We can describe algorithms in terms of fold,
unfold, and other recognised patterns.

– We can prove properties about folds,

– and apply the proved theorems to all programs
that are folds, either for compiler optimisa-
tion, or for mathematical reasoning.

• Among the theorems about folds, the most impor-
tant is probably the fold-fusion theorem.

The Fold-Fusion Theorem
The theorem is about when the composition of a func-

tion and a fold can be expressed as a fold.

Theorem 1 (foldr -Fusion). Given f :: a → b → b, e :: b,
h :: b → c, and g :: a → c → c, we have:

h · foldr f e = foldr g (h e) ,

if h (f x y) = g x (h y) for all x and y.

For program derivation, we are usually given h, f , and
e, from which we have to construct g.

Tracing an Example
Let us try to get an intuitive understand of the theo-

rem:

h (foldr f e [a, b, c])

= { definition of foldr }

h (f a (f b (f c e)))

= { since h (f x y) = g x (h y) }

g a (h (f b (f c e)))

= { since h (f x y) = g x (h y) }

g a (g b (h (f c e)))

= { since h (f x y) = g x (h y) }

g a (g b (g c (h e)))

= { definition of foldr }

foldr g (h e) [a, b, c] .

Sum of Squares, Again

• Consider sum · map square again. This time we
use the fact that map f = foldr (mf f) [ ], where
mf f x xs = f x : xs .

• sum ·map square is a fold, if we can find a ssq such
that sum (mf square x xs) = ssq x (sum xs). Let
us try:

sum (mf square x xs)

= { definition of mf }

sum (square x : xs)

= { definition of sum }

square x+ sum xs

= { let ssq x y = square x+ y }

ssq x (sum xs) .

Therefore, sum ·map square = foldr ssq 0.

Sum of Squares, without Folds
Recall that this is how we derived the inductive case

of sumsq yesterday:

sumsq (x : xs)

= { definition of sumsq }

sum (map square (x : xs))

= { definition of map }

sum (square x : map square xs)

= { definition of sum }

square x+ sum (map square xs)

= { definition of sumsq }

square x+ sumsq xs .

Comparing the two derivations, by using fold-fusion we
supply only the “important” part.

More on Folds and Fold-fusion

• Compare the proof with the one yesterday. They
are essentially the same proof.

• Fold-fusion theorem abstracts away the common
parts in this kind of inductive proofs, so that we
need to supply only the “important" parts.

Scan

• The following function scanr computes foldr for ev-
ery suffix of the given list:

scanr :: (a → b → b) → b → List a → List b
scanr f e = map (foldr f e) · tails .

• E.g. computing the running sum of a list:

scanr (+) 0 [8, 1, 3]
= map sum (tails [8, 1, 3])
= map sum [[8, 1, 3], [1, 3], [3], [ ]]
= [12, 4, 3, 0].

• Surely there is a quicker way to compute scanr ,
right?

Scan

• Recall that tails is a foldr :

tails = foldr (λx yss →
(x : head yss) : yss) [[ ]] .

• By foldr -fusion we get:

scanr f e = foldr (λx ys →
f x (head ys) : ys) [e] ,

• which is equivalent to this inductive definition:

scanr f e [ ] = [e]
scanr f e (x : xs) = f x (head ys) : ys ,

where ys = scanr f e xs .

Tupling as Fold-fusion

• Tupling can be seen as a kind of fold-fusion. The
derivation of steepsum, for example, can be seen
as fusing:

steepsum · id = steepsum · foldr (:) [ ].

• Recall that steepsum xs = (steep xs, sum xs). Re-
formulating steepsum into a fold allows us to com-
pute it in one traversal.

Accumulating Parameter as Fold-Fusion

• We also note that introducing an accumulating pa-
rameter can often be seen as fusing a higher-order
function with a foldr .

• Recall the function reverse . Observe that

reverse = foldr (λx xs → xs ++[x]) [ ] .

• Recall revcat xs ys = reverse xs ++ ys . It is equiv-
alent to

revcat = (++) · reverse .

• Deriving revcat is performing a fusion!

2 Folds on Other Algebraic
Datatypes

• Folds are a specialised form of induction.

• Inductive datatypes: types on which you can per-
form induction.

• Every inductive datatype give rise to its fold.

• In fact, an inductive type can be defined by its fold.

Fold on Natural Numbers

• Recall the definition:

data Nat = 0 | 1+ Nat .

• Constructors: 0 :: Nat , (1+) :: Nat → Nat .

• What is the fold on Nat?

foldN :: (a → a) → a → Nat → a
foldN f e 0 = e
foldN f e (1+ n) = f (foldN f e n) .

Examples of foldN

• (+n) = foldN (1+) n.

0 + n = n
(1+ m) + n = 1+ (m+ n) .

• (×n) = foldN (n+) 0.

0× n = 0
(1+ m)× n = n+ (m× n) .

• even = foldN not True .

even 0 = True
even (1+ n) = not (even n) .

Fold-Fusion for Natural Numbers

Theorem 2 (foldN -Fusion). Given f :: a → a, e :: a,
h :: a → b, and g :: b → b, we have:

h · foldN f e = foldN g (h e) ,

if h (f x) = g (h x) for all x.

Exercise: fuse even into (+)?

Folds on Trees

• Recall some datatypes for trees:

data ITree a = Null | Node a (ITree a) (ITree a) ,
data ETree a = Tip a | Bin (ETree a) (ETree a) .

• The fold for ITree , for example, is defined by:

foldIT :: (a → b → b → b) → b → ITree a → b
foldIT f e Null = e
foldIT f e (Node a t u) = f a (foldIT f e t) (foldIT f e u) .

• The fold for ETree , is given by:

foldET :: (b → b → b) → (a → b) → ETree a → b
foldET f k (Tip x) = k x
foldET f k (Bin t u) = f (foldET f k t) (foldET f k u) .

Some Simple Functions on Trees

• To compute the size of an ITree :

sizeITree = foldIT (λx m n → 1+ (m+ n)) 0 .

• To sum up labels in an ETree :

sumETree = foldET (+) id.

• To compute a list of all labels in an ITree and an
ETree :

flattenIT =foldIT (λx xs ys → xs ++[x] ++ ys) [ ],

flattenET =foldET (++) (λx → [x]).

• Exercise: what are the fusion theorems for foldIT
and foldET ?

3 Finally, Solving Maximum Seg-
ment Sum

Specifying Maximum Segment Sum

• Finally we have introduced enough concepts to
tackle the maximum segment sum problem!

• A segment can be seen as a prefix of a suffix.

• The function segs computes the list of all the seg-
ments.

segs = concat ·map inits · tails .

• Therefore, mss is specified by:

mss = max ·map sum · segs .

The Derivation!
We reason:

max ·map sum · concat ·map inits · tails
= { since map f · concat = concat ·map (map f) }

max · concat ·map (map sum) ·map inits · tails
= { since max · concat = max ·map max }

max ·map max ·map (map sum) ·map inits · tails
= { since map f ·map g = map (f.g) }

max ·map (max ·map sum · inits) · tails .

Recall the definition scanr f e = map (foldr f e) · tails .
If we can transformmax ·map sum · inits into a fold, we
can turn the algorithm into a scanr , which has a faster
implementation.

Maximum Prefix Sum
Concentrate onmax ·map sum · inits (let ini x xss =

[ ] : map (x :) xss):

max ·map sum · inits
= { definition of init, ini x xss = [ ] : map (x :) xss }

max ·map sum · foldr ini [[ ]]

= { fold fusion, see below }

max · foldr zplus [0] .

The fold fusion works because:

map sum (ini x xss)

= map sum ([ ] : map (x :) xss)

= 0 : map (sum · (x :)) xss

= 0 : map (x+) (map sum xss) .

Define zplus x yss = 0 : map (x+) yss .

Maximum Prefix Sum, 2nd Fold Fusion
Concentrate on max ·map sum · inits :

max ·map sum · inits
= { definition of init, ini x xss = [ ] : map (x :) xss }

max ·map sum · foldr ini [[ ]]

= { fold fusion, zplus x xss = 0 : map (x+) xss }

max · foldr zplus [0]

= { fold fusion, let zmax x y = 0 ↑ (x+ y) }

foldr zmax 0 .

The fold fusion works because ↑ distributes into (+):

max (0 : map (x+) xs)

=0 ↑ max (map (x+) xs)

=0 ↑ (x+max xs) .

Back to Maximum Segment Sum
We reason:

max ·map sum · concat ·map inits · tails
= { since map f · concat = concat ·map (map f) }

max · concat ·map (map sum) ·map inits · tails
= { since max · concat = max ·map max }

max ·map max ·map (map sum) ·map inits · tails
= { since map f ·map g = map (f.g) }

max ·map (max ·map sum · inits) · tails
= { reasoning in the previous slides }

max ·map (foldr zmax 0) · tails
= { introducing scanr }

max · scanr zmax 0 .

Maximum Segment Sum in Linear Time!

• We have derivedmss = max ·scanr zmax 0, where
zmax x y = 0 ↑ (x+ y).

• The algorithm runs in linear time, but takes linear
space.

• A tupling transformation eliminates the need for
linear space.

mss = fst ·maxhd · scanr zmax 0

where maxhd xs = (max xs, head xs). We omit
this last step in the lecture.

• The final program is mss = fst · foldr step (0, 0),
where step x (m, y) = ((0 ↑ (x + y)) ↑ m, 0 ↑
(x+ y)).

6


